TEMPLATE = lib
CONFIG += shared sockets
TARGET = xpinn
TARGET_EXT = .xpl

load(common)

SOURCES += *.cpp
HEADERS += *.h *.inc

DEFINES += ENABLE_DEBUG_EXPORTS

win32:LIBS += -lOpenGL32
else:LIBS += -lGL

INCLUDEPATH += $${LIBVATCONN}/include
CONFIG(debug,	debug|release):	LIBS += -L$${LIBVATCONN}/out/debug
CONFIG(release,	debug|release):	LIBS += -L$${LIBVATCONN}/out/release
LIBS += -llibvatconn
PRE_TARGETDEPS += $$decorateStaticLib(libvatconn)

isEmpty(XPSDK):error(XPSDK variable must point to the X-Plane SDK directory)

INCLUDEPATH += $${XPSDK}/CHeaders/XPLM $${XPSDK}/CHeaders/Widgets
win32:LIBS += -L$${XPSDK}/Libraries/Win -lXPLM -lXPWidgets
DEFINES += XPLM200=1 XPLM210=1